
C A S E S T U D Y

Printing a Chart for Each Customer
In this case study, you use a pivot table and a pivot chart to summarize data from a pivot table.The goal is to create a

pivot table with customers in the filter field area.The pivot table should summarize revenue by year and by product.

After this pivot table is created, you add a chart with a chart title.You can then loop through each pivot item in the

customer field. For each chart, you update the pivot table, change the chart title, and then export a chart with the name

of the customer.

There is a bit more complexity in this macro than in the others shown so far in this chapter.The process of grouping daily

dates up to yearly dates requires you to calculate the pivot table and then select one of the date fields. From

TableRange1, you have to select a cell that is one row and one column from the top-left corner before you can group

up to years.

After the first chart is created, you can loop through each pivot item in the customer pivot field. For each customer, you

change the page field, recalculate the pivot table, change the title in the chart, and then export.To do all this, you use the

following macro:

Sub CreateChartPerCustomer()

Dim WSD As Worksheet

Dim PTCache As PivotCache

Dim PT As PivotTable

Dim PRange As Range

Dim FinalRow As Long

Dim ChartDataRange As Range

Dim Cht As Chart

Dim PI As PivotItem

Set WSD = Worksheets(“Data”)

‘ Delete any prior pivot tables

For Each PT In WSD.PivotTables

PT.TableRange2.Clear

Next PT

WSD.Range(“M1:Z1”).EntireColumn.Clear

‘ Define input area and set up a Pivot Cache

FinalRow = WSD.Cells(Application.Rows.Count, 1).End(xlUp).Row

FinalCol = WSD.Cells(1, Application.Columns.Count). _

End(xlToLeft).Column

Set PRange = WSD.Cells(1, 1).Resize(FinalRow, FinalCol)

Set PTCache = ActiveWorkbook.PivotCaches.Create(SourceType:= _

xlDatabase, SourceData:=PRange.Address)

‘ Create the Pivot Table from the Pivot Cache

Set PT = PTCache.CreatePivotTable(TableDestination:=WSD. _

Cells(2, FinalCol + 2), TableName:=”PivotTable1”)

‘ Turn off updating while building the table

PT.ManualUpdate = True

Chapter 11 Creating Charts2

‘ Set up the row fields

PT.AddFields RowFields:=”Product”, ColumnFields:=”Date”, _

PageFields:=”Customer”

‘ Set up the data fields

With PT.PivotFields(“Revenue”)

.Orientation = xlDataField

.Function = xlSum

.Position = 1

End With

With PT

.ColumnGrand = False

.RowGrand = False

.NullString = “0”

End With

‘ Calc the pivot table

PT.ManualUpdate = False

PT.ManualUpdate = True

‘ Find a date field & group by year

PT.TableRange1.Offset(1, 1).Resize(1, 1).Group Start:=True, End:=True, _

Periods:=Array(False, False, False, False, False, False, True)

‘ Calc the pivot table

PT.ManualUpdate = False

PT.ManualUpdate = True

‘ CreateChart

Set ChartDataRange = _

PT.TableRange1.Offset(1, 0).Resize(PT.TableRange1.Rows.Count - 1)

‘ Add the Chart

WSD.Shapes.AddChart.Select

Set Cht = ActiveChart

Cht.SetSourceData Source:=ChartDataRange

‘ Format the Chart

Cht.ChartType = xlColumnClustered

Cht.SetElement (msoElementChartTitleAboveChart)

Cht.ChartTitle.Caption = “All Customers”

Cht.SetElement msoElementPrimaryValueAxisThousands

‘ Loop through each customer

For Each PI In PT.PivotFields(“Customer”).PivotItems

PT.PivotFields(“Customer”).ClearAllFilters

PT.PivotFields(“Customer”).CurrentPage = PI.Name

‘ Calc the pivot table

PT.ManualUpdate = False

PT.ManualUpdate = True

Cht.ChartTitle.Caption = PI.Name

Cht.Export Filename:=”C:\” & PI.Name & “.jpg”, Filtername:=”JPG”

Next PI

End Sub

Figure 11.22 shows one of the three dozen charts created by this macro in less than a minute.

Figure 11.22
The macro creates
dozens of summary
charts in less than
a minute.

